AFGL-TR-80-0004

LEVEL II

ADA083021

CHARACTERIZATION OF TITAN III-D ACOUSTIC PRESSURE SPECTRA BY LEAST-SQUARES FIT TO THEORETICAL MODEL

Eugene B. Hartnett
Eric Carleen

Boston College
Chestnut Hill, MA 02167

January 1980

Scientific Report No. 2

Approved for public release; distribution unlimited

DDC FILE COPY

AIR FORCE GEOPHYSICS LABORATORY
AIR FORCE SYSTEMS COMMAND
UNITED STATES AIR FORCE
HANSCOM AFB, MASS. 01731

80 4 14 010

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: AFGL-TR-80-0004	2. GOVT ACCESSION NO.
3. RECIPIENT'S CATALOG NUMBER: Interim rept.	
4. TITLE (and Subtitle): CHARACTERIZATION OF TITAN III-D ACOUSTIC PRESSURE SPECTRA BY LEAST-SQUARES FIT TO THEORETICAL MODEL	5. TYPE OF REPORT & PERIOD COVERED: Scientific, Interim Scientific Report No. 2
	6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s): Eugene B. Hartnett, Eric Carleen	8. CONTRACT OR GRANT NUMBER(s): F19628-78-C-0008
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Boston College, Chestnut Hill, MA 02167	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: 62010F, 7600-09-02
11. CONTROLLING OFFICE NAME AND ADDRESS: Air Force Geophysics Laboratory, Hanscom Air Force Base, MA 01731, Contract Manager: Henry A. Ossing/LWH	12. REPORT DATE: January 1980
	13. NUMBER OF PAGES: 58
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office): 59	15. SECURITY CLASS. (of this report): Unclassified
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release, Distribution unlimited

17 DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

rocket plume acoustics, power spectral density, least squares fit

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A theoretical model for the acoustic spectra of undeflected rocket plumes is fitted to computed spectra of a Titan III-D at varying times after ignition, by a least-squares method. Tests for the goodness of the fit are made.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

061 150

TABLE OF FIGURES

INTRODUCTION

Boston College, in its investigation of the seismic effects of rocket launchings on structures in the immediate environment, is analyzing data taken during a Titan III-D launch at Vandenberg AFB in March 1979. The work reported herein deals with the spectral form of the induced surface pressure. A theoretical model is fitted to the power spectral density of the observed pressure at various times during launch. The results are presented along with tests of their validity.

CHARACTERIZING PRESSURE SPECTRA

The data are contained on a file (LAUNC2.MNR) which begins 1.64 seconds before ignition and continues for 61.44 seconds at 100 samples per second on each of 16 channels. The array of instruments corresponding to these channels is shown in Fig. 1. For this report only the pressure sensor on channel 10 which is 950 feet from the launch point was used.

To obtain estimates of the power spectral density at a given time after ignition 256 samples were read in beginning at that time. These samples had been digitized at 204.8 counts per volt. The power spectra were computed by the periodogram technique employing on FFT algorithm. This periodogram is such that $\frac{1}{N}\sum_{i=1}^{N} x_i^2 = \sum_{k=1}^{N} S_k$.

It was found that the noise caused by the system and ambient pressure fluctuations was white and assumed to be additive above the quantization level. It was estimated by averaging 20 periodograms starting 1 minute before launch and found to average $9.37(10^{-8})$ volts^2 per cell (Fig. 2). This was subtracted from the rocket spectra to yield our best estimate of rocket induced surface pressure observed through our system. To convert these spectra to the spectra of the pressure appearing at the input, they were divided by the squared magnitude of the system response (Fig. 3). That is $S_k^{in} = \frac{S_k^{out}}{H_k^2}$. These were scaled by $1/\Delta f$ at 0hz and the Nyquist frequency and $2/\Delta f$ elsewhere so the resultant

power spectral density when integrated by trapezoidal rule from 0hz to the Nyquist frequency equals the mean square of the input. However since the value at 0hz is dependent on amplifier drift its value is discarded and to reduce contamination from errors in our estimate of system behavior near the Nyquist frequency only frequencies less than 40hz were considered.

From physical considerations (1) and experimental studies (2) it is believed that the power spectral density of the surface pressure caused by undeflected chemical rocket plumes is of the form:

$$P(w) = \frac{4}{\pi} \frac{P_{max}}{W_0} \left\{ \frac{W}{W_0} + \frac{W_0}{W} \right\}^{-2}$$

$$\text{or} \quad P(f) = \frac{4}{\pi} \frac{P_0}{f_0} \left\{ \frac{f}{f_0} + \frac{f_0}{f} \right\}^{-2}$$

It was desired to obtain values of P_0 and f_0 which minimized the sum of the squared differences between the observed power spectral density and the theoretical as described by the above equation. To find the sum of squared errors over the range of the power spectral density:

$$E = \sum_{k=1}^{129} (P'_k - P(f_k))^2$$

where P'_k is our estimate of the k^{th} value of the power spectral density and $f_k = 100(k-1)/256$ which is the frequency represented by the k^{th} values of the power spectral density.

For reasons already stated the full range of the power spectral density was not to be used so the limits of the

summation were changed accordingly.

$$E=\sum_{k=2}^{103}(P_k'-P(f_k))^2$$

Rather than finding both P_0 and f_0 by trial and error the derivative of E with respect to P_0 is taken and set to zero that is where the extremal would be found.

$$\frac{dE}{dP_0}=0=\sum_{k=2}^{103}2\left(P_k'-\frac{4P_0}{\pi f_0}\left\{\frac{f_k}{f_0}+\frac{f_0}{f_k}\right\}^{-2}\right)\left(-\frac{4}{\pi}f_0\left\{\frac{f_k}{f_0}+\frac{f_0}{f_k}\right\}^{-2}\right)$$

$$P_0=\frac{\pi f_0}{4}\,\frac{\sum\limits_{k=2}^{103}P_k'\left(\frac{f_k}{f_0}+\frac{f_0}{f_k}\right)^{-2}}{\sum\limits_{k=2}^{103}\left(\frac{f_k}{f_0}+\frac{f_0}{f_k}\right)^{-4}}$$

So for any value of f_0 the value of P_0 which insures minimum error in the least square sense is uniquely determined.

To find the value of f_0 which yields the smallest E (the 'best' f_0) an iteration scheme was devised. The initial search interval has f_2 and f_{103} as its endpoints in the belief that f_0 lies between them. Five trial values of f_0 are considered starting with the low endpoint of the search interval and increasing in equal steps to the high endpoint. For each trial value of f_0, P_0 and E are computed by the formulae above and the best f_0 selected. A new search interval is defined with endpoints equal to the trial values adjacent to the best f_0 during previous search and a new estimate of the best f_0 found. The process is repeated until f_0 is determined to within the limits of the computer's accuracy. In this case single precision yields about seven significant digits.

The resultant f_0 and P_0 were used to generate plots which show power spectral density based on observed values and theoretical values vs. normalized frequency, f/f_0. Table I shows f_0 and P_0 for each segment.

In order to compare average observed power with theoretical power the theoretical power spectral density was integrated from $-\infty$ to ∞.

$$P_t = \int_{-\infty}^{\infty} P(f)\,df = \frac{4}{\pi}\frac{P_0}{f_0}\int_{-\infty}^{\infty}\left(\frac{f}{f_0}+\frac{f_0}{f}\right)^{-2} df$$

$$= \frac{4}{\pi} P_0 f_0 \int_{-\infty}^{\infty} \frac{f^2}{(f^2+f_0^2)} df$$

$$= \frac{4P_0f_0}{\pi}\left\{ -\tfrac{1}{2}\frac{f}{(f^2+f_0^2)}\Big|_{-\infty}^{\infty} + \tfrac{1}{2}\int_{-\infty}^{\infty}\frac{1}{(f^2+f_0^2)} df \right. \tag{3}$$

$$= \frac{2P_0f_0}{\pi}\int_{-\infty}^{\infty}\frac{1}{(f^2+f_0^2)} df = \frac{2P_0f_0}{\pi}\left(\frac{\pi}{f_0}\right) = 2P_0 \tag{3}$$

The observed power $(\Delta\iota \sum_{k=1}^{9} P'_k)$ and theoretical power are contained in Table I.

VALIDATION

Although the theoretical curves do not appear to closely fit the observed spectra it is known that the use of the periodogram with a large number of samples to estimate the spectrum produces results which fluctuate wildly (4). It was desired to test whether these fluctuations fall within expected limits when the theoretical power spectral density

is assumed true.

Oppenheim and Schafer (4) show the development of an expression for the variance of the estimated P_k' 's for a white Gaussian process and Hinich and Clay (5) state that the result is a good approximation for a wide variety of random processes. They state that the variance (σ^2) of the spectrum is approximately equal to its magnitude squared.

$$\sigma_k^2 = \frac{1}{M} \sum_{m=1}^{M} (P_k'^{(m)} - \bar{P}_k')^2 = \bar{P}_k'^2$$

$$\text{or } \frac{\sigma_k^2}{P_k'^2} = \frac{1}{M} \sum_{m=1}^{M} \frac{(P_k'^{(m)} - \bar{P}_k')^2}{\bar{P}_k'^2}$$

A figure of merit is defined as $\frac{1}{102} \sum_{k=2}^{103} \frac{\sigma_k^2}{P_k'^2}$

which with M=1 and $\bar{P}_k'$ assumed equal to $P(f_k)$ becomes

$$\frac{1}{102} \sum_{k=2}^{103} \frac{(P_k' - P(f_k))^2}{P^2(f_k)}$$

This quantity which should be approximately 1 is contained in Table I. In addition as has been stated in (5), for large N $2\frac{P_k'}{P(f_k)}$ follows a chi-squared distribution with 2 degrees of freedom.

As an additional test of the validity of the fit this criteria was used. For each value of $A_k = \frac{2P_k'}{P(f_k)}$ its cumulative relative frequency was computed.

a_x = estimated probability that $A_k \le x$.

The chi-squared distribution for 2 degrees of freedom is

$$a=\frac{1}{2}\int_0^{\mu} e^{-\frac{x}{2}}dx=1-e^{-\frac{\mu}{2}} \quad (6).$$

Therefore $\mu=-2\ln(1-a)$

and $\mu_x=-2\ln(1-a_x)$

For a good fit μ_x should be approximately equal to x. x has been plotted vs. μ_x for each segment.

RESULTS

Results are presented in Table I and plots which are figure 4 through 12. Plots (a) are of the observed spectra (points only) and fitted theoretical curves (solid lines). Plots (b) are A_k vs. μ_x as previously defined and labelled 'observed test statistic' and 'theoretical test statistic' respectively. The vertical axis is scaled to fit the maximum value of A_k although this point cannot be plotted since its corresponding theoretical value is infinitely large. The horizontal axis stops at 9.25 which is the 99th percentile for the 2 degree of freedom chi-squared distribution. The solid line represents $A_k=\mu_x$. When the observed points lie above the line it means we had larger values than we should have expected. When they lie below it means they are smaller than expected.

CONCLUSIONS

In order to determine bounds on the figure of merit for acceptable fits simulated data was used. A random number

subroutine was used to produce normally distributed variables. For each group of 256 numbers the FFT was taken and the magnitude multiplied by the square root of P(f) for a set value of P_0 and f_0. The inverse transform was taken and the result multiplied by a decaying exponential which modelled the envelope of the rocket data. Two hundred similarly produced groups of 256 variables were fitted with the theoretical power spectral density curve and statistics of the figures of merit accumulated.

The figures of merit for this simulation were found to have a mean of 1.07 and a standard deviation of .35. The minimum value obtained was .57 and the maximum was 2.98. It was decided to accept the fitted data if its figure of merit fell within these extrema.

The first fit, which began 3.83 seconds after ignition, produced a figure of merit of 214.6. This fit is rejected. It is believed that this early in the launch the plume was not undeflected and the theoretical curve does not apply. Figures of merit for subsequent fits fell within extrema criterion and these fits are accepted.

It was found that when f_o was low (equal to 2.8 hz) the chi-squared test plot, while a straight line, lay above the $A_k=\mu_x$ line, as happened on a number of the fitted segments, and the estimated value of f_o was high (3.2hz). Evidently out estimates while acceptable according to the figure of merit criterion are biased toward the high frequencies.

REFERENCES

1. A. Powell, Theory of Vortex Sound, Journal of the Acoustical Society of America, Vol. 36, No. 1, Jan. 1964.

2. Acoustic Loads Generated by the Propulsion System, NASA SP-8072 June 1971.

3. R.S. Burington, Handbook of Mathematical Tables and Formulas, Fifth Edition, McGraw-Hill Book Company, New York, 1973.

4. Alan V. Oppenheim and Ronald W. Schafer, Digital Signal Processing, Prentice-Hall, Englewood Cliffs, N.J., 1975.

5. M.J. Hinich and C.S. Clay, The Application of the Discrete Fourier Transform in the Estimation of Power Spectra, Coherence, and Bispectra of Geophysical Data, Reviews of Geophysics, Vol. 6, No. 3, Aug. 1968.

6. D.B. Owen, Handbook of Statistical Tables, Addison-Wesley Publishing Co., Inc., Reading, Mass. 1962.

TABLE I

Time after Ignition (seconds)	f_o (hz)	P_o (psi^2)	$E((\frac{psi^2}{hz})^2)$	Observed Power (psi^2)	Theoretical Power (psi^2)	Figure of Merit
3.83	6.798	$8.477(10^{-5})$	$4.022(10^{-10})$	$7.931(10^{-5})$	$1.689(10^{-4})$	214.6
5.39	14.299	$6.741(10^{-4})$	$1.268(10^{-8})$	$3.875(10^{-4})$	$1.348(10^{-3})$	1.005
6.95	12.071	$1.606(10^{-3})$	$5.342(10^{-8})$	$1.007(10^{-3})$	$3.213(10^{-3})$	.620
8.51	11.190	$1.257(10^{-3})$	$6.848(10^{-8})$	$8.264(10^{-4})$	$2.514(10^{-3})$	1.381
10.07	7.548	$6.361(10^{-4})$	$1.648(10^{-8})$	$4.993(10^{-4})$	$1.272(10^{-3})$	1.528
11.63	5.235	$2.845(10^{-4})$	$5.418(10^{-9})$	$2.572(10^{-4})$	$5.690(10^{-4})$	1.813
13.19	6.097	$1.115(10^{-4})$	$7.147(10^{-10})$	$1.035(10^{-4})$	$2.230(10^{-4})$	1.405
14.75	4.390	$4.608(10^{-5})$	$2.198(10^{-10})$	$4.872(10^{-5})$	$9.216(10^{-5})$	2.59
16.31	2.855	$1.860(10^{-5})$	$9.203(10^{-11})$	$2.420(10^{-5})$	$3.719(10^{-5})$	1.032

VANDENBERG ARRAY



O PRESSURE SENSORS (9)

△ SEISMOMETERS (7)

Fig. 1


PERIODOGRAM OF AMBIENT NOISE
BEFORE LAUNCH (AVERAGE)
0.1180E-06
0.7275E-07
0.5900E-07
0.4425E-07
0.2950E-07
0.1475E-07
15.00
35.00
40.00
FREQUENCY (HZ)

Fig. 2


CHANNEL 10 RESPONSE MAGNITUDE VOLTS/PSI
17.19
15.47
13.75
12.03
10.31
8.596
5.157
1.719
5.000
45.00
FREQUENCY (HZ)

Fig. 3


CHANNEL 1
SECONDS AFTER IGNITION
1.E-06
1.E-07
PSI**2/HZ
.01
.1
1.
10.
100.
1.E-06
1.E-07
1.E-03
PSI**2/HZ
NORMALIZED FREQUENCY
Fig. 4a


SECONDS AFTER IGNITION
246.75
197.40
148.05
98.70
49.35
OBSERVED TEST STATISTICS
1.32
2.64
5.29
6.61
9.25
THEORETICAL TEST STATISTICS

Fig. 4b


NORMALIZED FREQUENCY
PSI**2/HZ
PSI**2/HZ
.00001
1.E-06
1.E-07
1.E-08
.00001
1.E-05
1.E-07
.01
.1
1.
10.
100.

Fig. 5a


CHANNEL 1C
5.?9 SECONDS AFTER IGNITION
OBSERVED TEST STATISTICS
9.25
7.71
6.17
4.62
3.08
1.54
0.00
1.32
2.64
3.96
5.29
6.61
7.93
THEORETICAL TEST STATISTICS

Fig. 5b


CHANNEL 10
SECONDS AFTER IGNITION
.0001
.00001
1.E-06
1.E-07
1.E-08
PSI**2/HZ
.001
.0001
.00001
1.E-05
1.E-07
PSI**2/HZ
.01
.1
1.
10.
100.
NORMALIZED FREQUENCY

Fig. 6a


CHANNEL 10
5.95 SECONDS AFTER IGNITION
OBSERVED TEST STATISTICS
9.25
7.71
6.17
4.62
3.08
1.54
0.00
1.32
2.64
3.96
5.29
6.61
THEORETICAL TEST STATISTICS

Fig. 6b


CHANNEL 1?
8.51 SECONDS AFTER IGNITION
.0001
.00001
1.E-06
1.E-07
.0001
.00001
1.E-05
1.E-07
PSI**2/HZ
PSI**2/HZ
.1
1.
10.
100.
NORMALIZED FREQUENCY

Fig. 7a


CHANNEL 10
SECONDS AFTER IGNITION
OBSERVED TEST STATISTICS
THEORETICAL TEST STATISTICS
0.00
2.60
5.20
7.80
10.40
13.00
15.60
1.32
2.64
3.96
5.29
9.25

Fig. 7b



CHANNEL 13
18.07 SECONDS AFTER IGNITION
NORMALIZED FREQUENCY
.01
.1
1.
10.
100.
1.E-06
1.E-07
1.E-05
1.E-07
.00001
P S I * * 2 / H Z
P S I * * 2 / H Z

Fig. 8a


CHANNEL 13
10.87 SECONDS AFTER IGNITION
OBSERVED TEST STATISTICS
14.04
11.23
8.43
5.62
2.81
1.32
5.29
9.25
THEORETICAL TEST STATISTICS

Fig. 8b


SECONDS AFTER IGNITION
.00001
1.E-06
1.E-07
1.E-05
PSI**2/HZ
.1
1.
10.
100.
NORMALIZED FREQUENCY

Fig. 9a


CHANNEL 10
11.63 SECONDS AFTER IGNITION
15.96
13.30
10.64
7.98
5.32
2.66
0.00
OBSERVED TEST STATISTICS
1.32
2.64
THEORETICAL TEST STATISTICS

Fig. 9b


CHANNEL 1
13.19 SECONDS AFTER IGNITION
PSI**2/HZ
.00001
1.E-05
1.E-07
1.E-08
.01
.1
1.
10.
NORMALIZED FREQUENCY

Fig. 10a


CHANNEL 10
13.19 SECONDS AFTER IGNITION
9.25
7.71
6.17
4.62
3.08
1.54
2.64
OBSERVED TEST STATISTICS
THEORETICAL TEST STATISTICS

Fig. 10b


CHANNEL 14
14.75 SECONDS AFTER IGNITION
.0001
1.E-06
1.E-07
1.E-08
1.E-09
PSI**2/HZ
.01
.1
1.
10.
100.
NORMALIZED FREQUENCY

Fig. 11a



THEORETICAL TEST STATISTICS

Fig. 11b



CHANNEL 19
16.31 SECONDS AFTER IGNITION
.00001
1.E-06
1.E-07
1.E-08
1.E-09
PSI**2/HZ
.1
1.
10.
100.
NORMALIZED FREQUENCY

Fig. 12a


CHANNEL 10
16.31 SECONDS AFTER IGNITION
OBSERVED TEST STATISTICS
22.33
18.61
14.89
11.16
7.44
3.72
1.32
2.64
3.96
5.29
6.61
7.93
9.25
THEORETICAL TEST STATISTICS

Fig. 12b

APPENDIX

Listing of computer program follows

```
      PROGRAM POWER4
C     THIS IS POWER2 WITH MUCH OF THE OUTPUT LEFT OUT.
C     LINK THIS WITH PLLIB3 FOR PLOTTING.
C     LINK WITH MISC FOR HISTOGRAM AND ASSOCIATED DATA TABLES,
C         DATMIN AND DATMAX FUNCTIONS, AND SEARCH SUBROUTINE.
C     LINK WITH FFTSUB FOR FAST FOURIER TRANSFORM SUBROUTINES.
C
C
      DIMENSION WORK(102), PTHEOR(300), ASCALE(9), IN(20), DUMMY(256),
     +              CONV(103), NSKIP(12), RESID(102), AJUNK(25), PMULT(4),
     +              STATS(102), THSTAT(102), DAT(3),VARPLT(1[illegible]),SSEPLT([illegible])
      LOGICAL*1 LABEL(6)
      COMMON PI, W(300), PPRIME(255)
      DATA PI/3.14159265358980/
      DATA ASCALE/175.9,131.4,204.8,171.4,203.9,153.9,152.9,204.,203.2/
      DATA NSKIP/0,235,391,547,703,859,1015,1171,1327,1483,1639,1795/
      DATA PMULT/.575,1.,2.,2.773/
C
      TYPE 10
10    FORMAT ( ' TYPE DATA FILE NAME:' /)
      CALL ASSIGN (1, ' ', -1, 'RDO')
      DEFINE FILE 1 (6144, 20, U, INDEX)
      CALL ASSIGN(2,'DK:VF2PRS.DAT',13,'RDO')
      DEFINE FILE 2 (257,6,U,JNDEX)
      TYPE 15
15    FORMAT(' TYPE FILE NAME FOR OUTPUT OF RESIDUALS AND THEORETICAL'
     +            / ' VALUES:' /)
      CALL ASSIGN(3,' ',-1)
      DEFINE FILE 3 (102,50,U,KNDEX)
C
      DELTAF = 100./256.
      RNOYZ = .935E-7
C
      JNDEX = 1
      DO 20 I=1,103
         READ (2'JNDEX)XXX,CONV(I),YYY
20       JNDEX = JNDEX + 1
C
      TYPE*,'CHANNEL NUMBER = ?'
      ACCEPT*,ICHAN
      TYPE*,'ISTART = ?'
      ACCEPT*,ISTART
C
      CALL DATE(DAT)
C
C
      DO 1000 ISKIP = ISTART,12
C
      PRINT 25, DAT
25    FORMAT(' PROGRAM POWER4'10X,3A4, //
     +            ' FILES USED:'13X'LAUNC2.MNR  INPUT'/
     +            25X'VF2PRS.DAT  INPUT' / 25X'DATA.LC2    OUTPUT' //)
      PRINT 30, ICHAN, NSKIP(ISKIP), RNOYZ/DELTAF
30    FORMAT('0*****CHANNEL', I3, 10X, I5, ' OBSERVATIONS SKIPPED.'
```

```
     +              ' NOISE SUBTRACTED =',1PE11.4)
C
C      READ IN NEW OBSERVED DATA
C
       INDEX = NSKIP(ISKIP) + 1
       DO 50 I=1,256
          DUMMY(I) = 0.
          READ (1'INDEX)IN
50        PPRIME(I) = IN(ICHAN)/ASCALE(ICHAN-7)
C
C      SUBROUTINE FFTFP CONVERTS ARRAY DATA INTO SPECTRUM
       CALL FFTFP (PPRIME,DUMMY,256,2,0)
C
C      CONVERT SPECTRUM INTO PSD (PPRIME ARRAY).  DROP FIRST POINT.
C      GENERATE FREQUENCY AND CONVERT TO OMEGA (W ARRAY).
C
C      THE END OF THE LOOP IS 102 INSTEAD OF 128, SO THAT THE MAXIMUM
C      FREQUENCY IS NOW 40 HZ, INSTEAD OF 50 HZ.
C
       PSUM = 0.
       DO 59 I=1,102
          PPRIME(I)=2.*(PPRIME(I+1)-RNOYZ)/(DELTAF*CONV(I+1)*CONV(I+1))
          IF(PPRIME(I) .LE. 0.)PRINT 58,FLOAT(I)*DELTAF,PPRIME(I)
58        FORMAT(' *****WARNING     AT', F5.1,' HZ  PSI**2/HZ =',G11.4)
          PSUM = PSUM + PPRIME(I)
59        W(I) = FLOAT(I) * DELTAF
C
       IF(ISTART .NE. ISKIP) GOTO 75
       KNDEX = 1
       DO 70 I=1,102
          READ (3'KNDEX)AJUNK
          AJUNK(25) = W(I)
          KNDEX = KNDEX - 1
70        WRITE(3'KNDEX)AJUNK
C
C      FIND THE TRIAL FREQUENCY THAT GENERATES THEORETICAL VALUES THAT
C      BEST MATCH THE OBSERVED VALUES (CRITERION: LEAST SUM OF SQUARED
C      ERRORS).  STORE ANSWER IN W0.  STORE ASSOCIATED BEST P IN P0.
C
75     CALL SEARCH(W0,SSE,4,0,W(1),W(102),0.,7,WORK,PPRIME,102,0,0.
     +              7.,5.,'TRIAL FREQUENCY',15,VARPLT,SSEPLT,100)
500    P0 = PNOT(W0)
C
C      USING THE ESTIMATES OF P AND W, CALCULATE THE CURVE (PTHEOR).
C      THEN CALCULATE THE RESIDUALS AND THE FIGURE OF MERIT (DOLLY)
       CALL FUNCT(W0,PTHEOR)
       RESBAR = 0.
       DOLLY = 0.
       DO 550 I=1,102
          RESID(I) = PPRIME(I) - PTHEOR(I)
          RESBAR = RESBAR + ABS(RESID(I))
550       DOLLY = DOLLY + (RESID(I)/PTHEOR(I))**2
       RESBAR = RESBAR/102.
       DOLLY = DOLLY/102.
```

```
      C
      C     PRINTOUT RESULTS
      C
            PRINT 575, W0, P0, SSE, DELTAF*PSUM, 2.*P0, RESBAR, DOLLY
      575   FORMAT (///' F-NOT ESTIMATE (HZ)', T32, 1PE15.7  /
           +        ' P ESTIMATE', T32, 1PE15.7   /
           +        ' SUM OF SQUARED ERRORS (SSE)', T32, 1PE15.7   /
           +        ' TOTAL OBSERVED POWER', T32, 1PE15.7    /
           +        ' TOTAL THEORETICAL POWER', T32, 1PE15.7    /
           +        ' AVERAGE ABSOLUTE RESIDUAL', T32, 1PE15.7  /
           +        ' FIGURE OF MERIT',T32,1PE15.7, //// )
      C
      C
      C     PRINTOUT RESIDUALS AND THEORETICAL VALUES WITH FREQUENCY.
      C     STORE THEM IN DATA FILE 3.
      C
      C     THE FOLLOWING PRINTOUT IS OPTIONAL
      CC    PRINT 593
      593   FORMAT('  FREQ    THEORETICAL', 2( 29X 'FREQ    THEORETICAL') /
           +   '  (HZ)      PSI**2/HZ' 5X 'RESIDUAL', 2(16X '(HZ)      PSI**2/HZ'
           +            5X 'RESIDUAL'))
      CC    DO 600 I=1,34
      CC600    PRINT 610,((W(J),PTHEOR(J),RESID(J)),J=I,(I+68),34)
      610   FORMAT(1X,F5.2, 2(5X,1PE9.2),0P, 2(15X, F5.2, 2(5X, 1PE9.2),0P))
            PRINT 625
      625   FORMAT('1')
            KNDEX = 1
            DO 650 I=1,102
               READ(3'KNDEX)AJUNK
               LDEX = ISKIP*2 - 1
               AJUNK(LDEX) = PTHEOR(I)
               AJUNK(LDEX + 1) = RESID(I)
               KNDEX = KNDEX - 1
      650      WRITE(3'KNDEX)AJUNK
      C
      C     OPTIONAL PLOT
      C
      C
      C     PLOT PPRIME(OBSERVED) VS PTHEOR(THEORETICAL)
      C     INCLUDE 4 COMPARISON LINES:
      C         1) ( .575 * PTHEOR) VS PTHEOR
      C         2) (1.000 * PTHEOR) VS PTHEOR
      C         3) (2.000 * PTHEOR) VS PTHEOR
      C         4) (2.773 * PTHEOR) VS PTHEOR
      C
      C     PLOTTING INFO:
            PTMIN = AMIN1(PTHEOR(1),PTHEOR(102))
            PPMIN = DATMIN(PPRIME,102)
            YMIN = AMIN1(PPMIN,.5*PTMIN)
            PTMAX = DATMAX(PTHEOR,102)
            PPMAX = DATMAX(PPRIME,102)
            YMAX = AMAX1(PPMAX,2.773*PTMAX)
            DELTAY = (YMAX - YMIN)/6.
            DELTAX = (PTMAX - PTMIN)/7.
```

```
      C
            GOTO 715 !PLOT IS SKIPPED.
      C
            CALL PLOTS(6)
            CALL PLOT(1.,-6., 3)
            CALL AXIS1(2.,0.,'OBSERVED PSI**2/HZ',18,6.,90.,YMIN,DELTAY)
            CALL AXIS1(2.,0.,'THEORETICAL PSI**2/HZ',-21,7.,0.,PTMIN,DELTAX)
            CALL AXIS1(7.,0.,'OBSERVED PSI**2/HZ',-18,6.,90.,YMIN,DELTAY)
            CALL LINE(PTHEOR,PTMIN,DELTAX,PPRIME,YMIN,DELTAY,102,1,-1,'*')
      C     PLOT THE 3 COMPARISON LINES
            DO 710 I=1,4
               CALL PLOT(0.,(PMULT(I)*PTMIN - YMIN)/DELTAY,3)
        710    CALL PLOT(7.,(PMULT(I)*PTMAX - YMIN)/DELTAY,2)
            PRINT 625
      C
      C     CALCULATE AND PLOT TEST STATISTICS VS FREQUENCY
      C
        715 DO 720 I=1,102
        720    STATS(I) = 2.*PPRIME(I)/PTHEOR(I)
      C
      C     OPTIONAL PLOT
      C
            GOTO 725
            CALL DOPLOT(7.,6.,W,STATS,102,'FREQUENCY (HZ)',14,
           1                'OBSERVED TEST STATISTICS',24,-1,'*',1,1)
            PRINT 625
      C
      C     OPTIONAL HISTOGRAM OF TEST STATISTICS
      C
      CC    CALL HIST(7.,6.,0,'OBSERVED TEST STATISTICS',24,STATS,102,0,1,0)
      C
      C     ORDER DATA (NOT NECESSARY IF HIST WAS USED):

        725 DO 735 I=1,101
               MIN = I
               DO 730 J=I,102
                IF(STATS(J) .LT. STATS(MIN)) MIN = J
        730     CONTINUE
               S = STATS(I)
               STATS(I) = STATS(MIN)
        735    STATS(MIN) = S
      C
      C
      C     CALCULATE THEORETICAL TEST STATISTICS BASED ON INVERSE OF
      C      CHI SQUARE DISTRIBUTION.
      C     THIS IS DONE FOR THE FIRST 101 OF THE 102 POINTS; THE
      C     THEORETICAL VALUE FOR THE LAST POINT IS INFINITE.
      C
            DO 740 I=1,101
               OBSPRB = I/102.
        740    THSTAT(I) = -2.*ALOG(1.-OBSPRB)
      C
      C     PLOT THEORETICAL VS OBSERVED VALUES.
      C
```

```
              DELTAY = AMAX1(STATS(102),THSTAT(101))/6.
              DELTAX = THSTAT(101)/7.
              CALL PLOTS(6)
              CALL RED
              CALL PLOT(1.,-6.5,-3)
              CALL AXIS1(Ø.,Ø.,'OBSERVED TEST STATISTICS',24,6.,9Ø.,Ø.,DELTAY)
              CALL AXIS1(Ø.,Ø.,'THEORETICAL TEST STATISTICS',-27,7.,Ø.,Ø.,
            +                DELTAX)
              CALL AXIS1(7.,Ø.,'OBSERVED TEST STATISTICS',-24,6.,9Ø.,Ø.,DELTAY)
              CALL LINE(THSTAT,Ø.,DELTAX,STATS,Ø.,DELTAY,1Ø1,1,-1,'*')
              CALL PLOT( THSTAT(1)/DELTAX , THSTAT(1)/DELTAY ,3)
              CALL PLOT( THSTAT(1Ø1)/DELTAX , THSTAT(1Ø1)/DELTAY ,2)
      C       NSKIP EQUALS SECONDS*1ØØ PASSED
              TIME = FLOAT(NSKIP(ISKIP))/1ØØ. - 1.64
              ENCODE (6,75Ø,LABEL(1))TIME
       75Ø    FORMAT (F6.2)
              CALL SYMBOL (1.,6.,Ø,'CHANNEL 1Ø',Ø.,1Ø)
              CALL SYMBOL (3.,6.,Ø,LABEL,Ø.,6)
              CALL SYMBOL (3.75,6.,Ø,'SECONDS AFTER IGNITION',Ø.,22)
              PRINT 625
      C
      C       THE FOLLOWING EXTENDS THE PTHEOR AND W ARRAYS SO THAT THE
      C       THEORETICAL CURVE IS SYMMETRIC
      C       NPOINT = THE NUMBER OF POINTS NECESSARY FOR A SYMMETRIC CURVE.
      C
              NPOINT = 1Ø1
              INDEX1 = 1Ø2
       775    NPOINT = NPOINT + 1
              W(NPOINT) = FLOAT(INDEX1)*DELTAF
              PTHEOR(NPOINT) = PCURVE(WØ,PØ,W(NPOINT))
              INDEX1 = INDEX1 + 1Ø
              IF(PTHEOR(NPOINT) .GT. PTHEOR(1)) GOTO 775
      C
      C       NORMALIZE W BY DIVIDING BY WØ
              DO 8ØØ I=1,NPOINT
       8ØØ       W(I) = W(I)/WØ
      C
      C       LOG-LOG PLOT OF BOTH OBSERVED AND THEORETICAL VALUES VS
      C       FREQUENCY.
      C
              PMIN = AMIN1(PPMIN,PTHEOR(1),PTHEOR(NPOINT))
              PMAX = AMAX1(PPMAX,PTMAX)
              CALL PLOTS(6)
              CALL RED
              CALL PLOT (1.,-6.5,-3)
              CALL LAXIS(Ø.,Ø.,'PSI**2/HZ',9,6.,9Ø.,PMAX,PMIN,PDMIN,PDELTA)
              CALL LAXIS(7.,Ø.,'PSI**2/HZ',-9,6.,9Ø.,PMAX,PMIN,PDMIN,PDELTA)
              CALL LAXIS(Ø.,Ø.,'NORMALIZED FREQUENCY',-2Ø,7.,Ø.,W(NPOINT),
            +                W(1),WDMIN,WDELTA)
              CALL LOGLOG (W,PTHEOR,NPOINT,1,WDMIN,WDELTA,PDMIN,PDELTA,Ø,'.')
              CALL LOGLOG (W,PPRIME,1Ø2,1,WDMIN,WDELTA,PDMIN,PDELTA,-1,'*')
              CALL SYMBOL (1.,6.2,Ø,'CHANNEL 1Ø',Ø.,1Ø)
              CALL SYMBOL (3.,6.2,Ø,LABEL,Ø.,6)
              CALL SYMBOL (3.75,6.2,Ø,'SECONDS AFTER IGNITION',Ø.,22)
```

```
            PRINT 625
      C
            GOTO 1000
      C
      C     CONVERT W INTO REGULAR FREQUENCY AND PLOT ABOVE CURVES ON LINEAR
      C     AXES.
      C
            DO 950 I=1,102
      950      W(I) = W(I)*W0
            WINC = (W(102) - W(1))/7.
            CALL PLOT (0.,-6.,-3)
            PMIN = AMIN1(PPMIN,PTMIN)
      C     PMAX IS THE SAME AS ABOVE, SO:
            PINC = (PMAX - PMIN)/6.
            CALL AXIS1 (0.,0.,'PSI**2/HZ',9,6.,90.,PMIN,PINC)
            CALL AXIS1 (7.,0.,'PSI**2/HZ',-9,6.,90.,PMIN,PINC)
            CALL AXIS1 (0.,0.,'FREQUENCY',-9,7.,0.,W(1),WINC)
            CALL LINE (W,W(1),WINC,PTHEOR,PMIN,PINC,102,1,0,'.')
            CALL LINE (W,W(1),WINC,PPRIME,PMIN,PINC,102,1,-1,'*')
            CALL SYMBOL (1.,6.2,0,'CHANNEL 10',0.,10)
            CALL SYMBOL (3.,6.2,0,LABEL,0.,6)
            CALL SYMBOL (3.75,6.2,0,'SECONDS AFTER IGNITION',0.,22)
            PRINT 625
      1000  CALL CLOSE(6)
      C
            STOP
            END
```

```
C
C
C     THIS FUNCTION RETURNS THE BEST P-NOT VALUE FOR A GIVEN
C     OMEGA-NOT.  CRITERION IS LEAST SSE.
C
      FUNCTION PNOT(WNOT)
      COMMON PI, W(300), PPRIME(256)
      DENSUM = 0.
      TOPSUM = 0.
      DO 100 I=1,102
         WSUM = W(I)/WNOT + WNOT/W(I)
         DENSUM = DENSUM + 1./(WSUM**4)
  100    TOPSUM = TOPSUM + PPRIME(I)/(WSUM*WSUM)
      PNOT = PI * WNOT * TOPSUM/(4. * DENSUM)
      RETURN
      END
```

```
FORTRAN IV      V02.04      Fri 09-Nov-79 00:00:00                 PAGE 001

      C
      C
      C       THIS FUNCTION RETURNS A P(I) VALUE FOR A GIVEN W(I)
      C       W(I) IN OMEGA FORM.
      C
              FUNCTION PCURVE(WNOT,P0,WI)
              COMMON PI
              WSUM = WI/WNOT + WNOT/WI
              PCURVE = 4. * P0/(PI * WNOT * WSUM * WSUM)
              RETURN
              END
```

```
FORTRAN IV      V02.04      Fri 09-Nov-79 00:00:00                 PAGE 001

      C
      C
              SUBROUTINE FUNCT(VAR,WORK)
              DIMENSION WORK(102)
              COMMON PI,W(300)
              P0 = PNOT(VAR)
              DO 100 I=1,102
      100        WORK(I) = PCURVE(VAR,P0,W(I))
              RETURN
              END
```

```
0001          SUBROUTINE FFTFP (XREAL,XIMAG,N,M,IF)
        C
        C     IF=0   FORWARD TRANSFORM
        C     IF=1   INVERSE TRANSFORM
        C
        C     M=0 XREAL AND XIMAG RETURNED AS REAL AND IMAG. FOR FORWARD XFORMS
        C     M=1    "     "     "      "        " MAGNITUDE AND PHASE "     "     "
        C                                         (PHASE IN DEGREES)
        C     M=2        XREAL RETURNED AS  'PSD'; XIMAG =0.
        C     HERE 'PSD' MEANS SUM OF N VALUES OF XREAL = MEAN SQUARE OF INPUT
        C
        C     FOR INVERSE TRANSFORM M DEFINITIONS APPLY TO INPUT DATA
        C     XREAL AND XIMAG RETURNED AS REAL AND IMAGINARY
        C
        C     FOR FORWARD TRANSFORMS XREAL AND XIMAG INPUT AS REAL AND IMAGINARY
        C
0002          DIMENSION XREAL(1),XIMAG(1)
0003          PI=3.14159
0004          DTOR=PI/180.
0005          IF(IF.EQ.0)GO TO 6
        C
        C     MUST PREPARE FOR INVERSE TRANSFORM
        C
0007          IF(M.EQ.0)GO TO 2
0009          IF(M.EQ.2)GO TO 4
        C
        C     INPUT IS MAGNITUDE AND PHASE
        C
0011          DO 1 I=1,N
0012          FMAG=XREAL(I)/N
0013          XREAL(I)=FMAG*COS(XIMAG(I)*DTOR)
0014          XIMAG(I)=-FMAG*SIN(XIMAG(I)*DTOR)
0015    1     CONTINUE
0016          GO TO 6
        C
        C     INPUT IS REAL AND IMAGINARY
        C
0017    2     DO 3 I=1,N
0018          XREAL(I)=XREAL(I)/N
0019          XIMAG(I)=-XIMAG(I)/N
0020    3     CONTINUE
0021          GO TO 6
        C
        C     INPUT IS 'PSD'
        C
0022    4     FACT=FLOAT(N)*N
0023          DO 5 I=1,N
0024          XREAL(I)=XREAL(I)/FACT
0025    5     CONTINUE
        C
0026    6     CALL FFTB (XREAL,XIMAG,N)
        C
0027          IF(IF.EQ.1)RETURN
        C
```

```
C       TRANSFORM WAS FORWARD
C
        IF(M.EQ.0)RETURN
        IF(M.EQ.2)GO TO 8
C
C       DESIRE OUTPUT IN MAGNITUDE AND PHASE
C
        DO 7 I=1,N
        XMAG=SQRT(XREAL(I)*XREAL(I)+XIMAG(I)*XIMAG(I))
        XIMAG(I)=ATAN2(XIMAG(I),XREAL(I))/DTOR
        XREAL(I)=XMAG
7       CONTINUE
        RETURN
C
C       DESIRE 'PSD'
C
8       FACT=FLOAT(N)*N
        DO 9 I=1,N
        XREAL(I)=(XREAL(I)*XREAL(I)+XIMAG(I)*XIMAG(I))/FACT
        XIMAG(I)=0.
9       CONTINUE
        RETURN
        END
```

```
      SUBROUTINE FFTB (XREAL,XIMAG,N)
      DIMENSION XREAL(N),XIMAG(N)
      NU=LOG2(N)
      N2=N/2
      NU1=NU-1
      K=0
      DO 100 L=1,NU
102   DO 101 I=1,N2
      P=IBITR(K/2**NU1,NU)
      ARG=6.283185*P/FLOAT(N)
      C=COS(ARG)
      S=SIN(ARG)
      K1=K+1
      K1N2=K1+N2
      TREAL=XREAL(K1N2)*C+XIMAG(K1N2)*S
      TIMAG=XIMAG(K1N2)*C-XREAL(K1N2)*S
      XREAL(K1N2)=XREAL(K1)-TREAL
      XIMAG(K1N2)=XIMAG(K1)-TIMAG
      XREAL(K1)=XREAL(K1)+TREAL
      XIMAG(K1)=XIMAG(K1)+TIMAG
101   K=K+1
      K=K+N2
      IF(K.LT.N)GO TO 102
      K=0
      NU1=NU1-1
100   N2=N2/2
      DO 103 K=1,N
      I=IBITR(K-1,NU)+1
      IF(I.LE.K)GO TO 103
      TREAL=XREAL(K)
      TIMAG=XIMAG(K)
      XREAL(K)=XREAL(I)
      XIMAG(K)=XIMAG(I)
      XREAL(I)=TREAL
      XIMAG(I)=TIMAG
103   CONTINUE
      RETURN
      END
```

```
      FUNCTION IBITR(J,NU)
      J1=J
      IBITR=0
      DO 200 I=1,NU
      J2=J1/2
      IBITR=IBITR*2+(J1-2*J2)
200   J1=J2
      RETURN
      END
```

```
0001          FUNCTION LOG2(N)
0002          N1=N
0003          J=1
0004          LOG2=0
0005    1     IF(J.EQ.N1)RETURN
0007          IF(J.GT.N1)GO TO 2
0009          J=J*2
0010          LOG2=LOG2+1
0011          GO TO 1
0012    2     TYPE 1000,N1
0013    1000  FORMAT (1X,I5,' IS NOT A POWER OF 2')
0014          STOP
0015          END
```

```
      SUBROUTINE HANN (RIN,N)
      DIMENSION RIN(1),TEMP(256)
      M=N-1
      DO 1 I=2,M
      TEMP(I)=(RIN(I-1)+RIN(I+1))/4+RIN(I)/2
    1 CONTINUE
      RIN(1)=(RIN(1)+RIN(2))/2
      RIN(N)=(RIN(N)+RIN(M))/2
      DO 2 I=2,M
      RIN(I)=TEMP(I)
    2 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE HAN2 (RIN,N)
      DIMENSION RIN(1),TEMP(256)
      M=N-1
      DO 1 I=2,M
      TEMP(I)=(RIN(I-1)+RIN(I+1))/3.+RIN(I)/3.
    1 CONTINUE
      RIN(1)=(RIN(1)+RIN(2)+RIN(3))/3
      RIN(N)=(RIN(N)+RIN(M)+RIN(M-1))/3
      DO 2 I=2,M
      RIN(I)=TEMP(I)
    2 CONTINUE
      RETURN
      END
```

```
C
C
C      SUBROUTINE SEARCH(THETA,SSETHE,AMPTHE,IAMP,ALOW,HIGH,ACCTHE,
C      IACCSE,WORK,STAN,NPTS,IPRINT,IPLOT,XSIZE,YSIZE,VARNAM,NCHAR,
C      VARPLT,SSEPLT,NPLTPT)
C
C      PURPOSE
C
C         DOES A "STAB" SEARCH FOR AN UNKNOWN PARAMETER.
C
C      ARGUMENTS
C
C         THETA
C          VALUE OF PARAMETER RETURNED AFTER THE SEARCH.
C
C         SSETHE
C          SSE ASSOCIATED WITH THETA, RETURNED BY PROGRAM.
C
C         AMPTHE
C          AMPLITUDE RATIO FOUND AND RETURNED.  IF NO RATIO
C          CALCULATIONS ARE DESIRED, SET IAMP = 0
C
C         IAMP
C          SEE ABOVE.
C
C         ALOW
C          USER-SUPPLIED.  LOW END OF RANGE IN WHICH THETA IS
C          EXPECTED TO APPEAR.
C
C         HIGH
C          USER-SUPPLIED.  HIGH END OF RANGE FOR THETA.
C          IF THE MINIMUM SSE IS FOUND AT EITHER ALOW OR HIGH,
C          THE PROGRAM WILL EXPAND THE RANGE UNTIL A NON-BOUNDARY
C          MINIMUM SSE IS FOUND.
C
C         ACCTHE
C          USER-SUPPLIED.  THE SEARCH ENDS WHEN EITHER THE
C          PARAMETER OR ITS SSE HAS BEEN FOUND TO SPECIFIED
C          PRECISIONS.
C
C          WHEN THE DIFFERENCE BETWEEN SUCCESSIVE ESTIMATES OF THE
C          PARAMETER IS LESS THAN OR EQUAL TO ACCTHE, THE SEARCH
C          IS STOPPED.
C
C          SETTING ACCTHE = 0. RESULTS IN THE COMPUTER SEARCHING TO
C          THE LIMITS OF ITS OWN ACCURACY.
C
C         IACCSE
C          WHEN SUCCESSIVE SSE'S AGREE TO IACCSE DIGITS THE SEARCH
C          IS STOPPED.  MAXIMUM = 7
C
C         WORK
C          WORK VECTOR.  LENGTH = NPTS.
C
```

```
      C        STAN
      C         USER-SUPPLIED DATA VECTOR CONTAINING THE POINTS THE
      C         PROGRAM TRIES TO MATCH.
      C
      C        NPTS
      C         NUMBER OF POINTS IN STAN.
      C
      C        IPRINT
      C         UNLESS SET = 0, USER GETS PRINTOUT OF INTERMEDIATE
      C         RESULTS OF THE SEARCH.
      C
      C        IPLOT
      C         PLOTTING CONTROL CHARACTER
      C          =0     NO PLOT
      C          =1     PLOT WITH REGULAR AXES
      C          =2     SEMILOG PLOT (SSE'S ON LOG AXIS)
      C          =3     BOTH PLOTS
      C
      C        XSIZE
      C         SIZE OF X-AXIS OF THE OPTIONAL PLOT.
      C
      C        YSIZE
      C         SIZE OF Y-AXIS OF THE OPTIONAL PLOT.
      C
      C        VARNAM
      C         HOLLERITH STRING NAME OF PARAMETER.
      C
      C        NCHAR
      C         NUMBER OF CHARACTERS IN VARNAM.
      C
      C        VARPLT
      C         ARRAY USED FOR PLOTTING.  LENGTH = NPTPLT
      C         UPON RETURN CONTAINS ALL ESTIMATES OF THE VARIABLE,
      C         ORDERED SMALL TO LARGE.
      C
      C        SSEPLT
      C         ARRAY USED FOR PLOTTING.  LENGTH = NPTPLT
      C         UPON RETURN CONTAINS SSE FOR EACH VARIABLE ESTIMATE
      C
      C        NPTPLT
      C         SIZE OF PLOTTING ARRAYS.
      C         IF THE INITIAL RANGE (HIGH - ALOW) HOLDS THE BEST VALUE
      C         OF THE UNKNOWN VARIABLE, THEN THE FOLLOWING FORMULA SHOULD
      C         PROVIDE AN UPPER BOUND ON NPTPLT:
      C
      C           NPTPLT <= 3 + 2*(LOG(RANGE/ACCTHE))/LOG(2)
      C
      C      THE USER MUST WRITE A SUBROUTINE FUNCT(VAR,WORK) THAT GENERATES
      C      A VECTOR "WORK" OF ESTIMATED DATA POINTS, GIVEN THAT THE UNKNOWN
      C      PARAMETER EQUALS "VAR".
      C
0001         SUBROUTINE SEARCH(THETA,SSETHE,AMPTHE,IAMP,ALOW,HIGH,ACCTHE,
     +          IACCSE,WORK,STAN,NPTS,IPRINT,IPLOT,XSIZE,YSIZE,VARNAM,NCHAR,
     +          VARPLT,SSEPLT,NPLTPT)
```

```
      DIMENSION VAR(5),SSE(5),AMP(5),WORK(NPTS),STAN(NPTS),
     1          VARPLT(NPLTPT),SSEPLT(NPLTPT)
      LOGICAL*1 ACCSE1(14),ACCSE2(14),VARNAM(NCHAR),ALPH(27)
      DATA ALPH/' ','A','B','C','D','E','F','G','H','I','J','K',
     1  'L','M','N','O','P','Q','R','S','T','U','V','W','X','Y','Z'/
C
      IF(IACCSE .GT. 7) STOP 'SSE ACCURACY CONSTANT MUST BE <= 7'
C
      DIFOLD = 0. ! USED IN PRECISION CHECK SECTION
      IROUND = 1  ! MARKS THE ROUND JUST COMPLETED
      NPOINT = 0  ! MARKS THE NUMBER OF POINTS ENTERED INTO
C                   THE PLOTTING ARRAYS
      ISTOP = 0   ! IF THE PLOTTING ARRAYS ARE FILLED, SSESUB
C                   SUBROUTINE SETS THIS EQUAL TO 1, WHICH IN
C                   TURN CAUSES THE SEARCH TO STOP.
      NOTE = 1    ! INDEX OF THE "ALPH" ARRAY.  USED WHEN THE
C                   SEARCH RANGE IS EXTENDED.
      ADD = (HIGH - ALOW)*.25
C
C     SET UP INITIAL VAR(1), VAR(3), AND VAR(5) WITH SSE'S
C
      VAR(5) = HIGH
      CALL SSESUB(VAR(5),SSE(5),AMP(5),IAMP,WORK,STAN,NPTS,IPLOT,
     1          NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
C
      VAR(3) = ALOW + .5*(HIGH-ALOW)
      CALL SSESUB(VAR(3),SSE(3),AMP(3),IAMP,WORK,STAN,NPTS,IPLOT,
     1          NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
C
  100 VAR(1) = VAR(3) - (VAR(5) - VAR(3))
      IF(VAR(3) .NE. VAR(1)) GOTO 125     !PRECISION CHECK REQUIRED BECAUS
C                                           RANGE EXTENSION SENDS CONTROL
C                                           BACK HERE.
      ISTART = 2
      GOTO 434
  125 CALL SSESUB(VAR(1),SSE(1),AMP(1),IAMP,WORK,STAN,NPTS,IPLOT,
     1          NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
C
C
C     SET UP VAR(2) WITH SSE(2)
C
  150 VAR(2) = (VAR(3) - VAR(1))*.5 + VAR(1)
      CALL SSESUB(VAR(2),SSE(2),AMP(2),IAMP,WORK,STAN,NPTS,IPLOT,
     1          NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
C
C     IF EITHER OF SSE(1) OR SSE(2) IS THE MINIMUM, THEN THERE IS NO
C     NEED TO FIND SSE(4).  IF SSE(1) IS MIN, EXTEND THE RANGE DOWNWARD.
C     IF SSE(2) IS MIN, GO ON TO PRECISION CHECKS BEFORE DOING ANOTHER
C     ROUND.  OTHERWISE, COMPUTE VALUES FOR VAR(4).
C
      IF((SSE(1) .GT. SSE(2)) .OR. (SSE(1) .GT. SSE(3)) .OR.
     1          (SSE(1) .GT. SSE(5))) GOTO 200
      IF(IPRINT .EQ. 0) GOTO 170
      PRINT 160, IROUND,ALPH(NOTE),VAR,SSE
```

```
      160   FORMAT(' ROUND',I3,A1 / ' VARIABLE:' 5G15.5 / ' SSE:'5X, 5G15.5)
            NOTE = NOTE + 1
            IF(NOTE .EQ. 28) NOTE = 27
            CALL CLOSE(6)
      170   IF(ISTOP .NE. 0) GOTO 650
      C     RESET ARRAYS AND DO ANOTHER ROUND:
            CALL RESET(VAR,1,3,4,5)
            CALL RESET(SSE,1,3,4,5)
            CALL RESET(AMP,1,3,4,5)
            GOTO 100
      C
      200   IF((SSE(2) .GT. SSE(3)) .OR. (SSE(2) .GT. SSE(5))) GOTO 300
      C     HERE IT IS KNOWN THAT VAR(2) IS THE BEST VALUE SO FAR
            ISTART = 1
            GOTO 420
      C
      C     SET UP VAR(4) WITH SSE(4)
      C
      300   VAR(4) = (VAR(5) - VAR(3))*.5 + VAR(3)
            CALL SSESUB(VAR(4),SSE(4),AMP(4),IAMP,WORK,STAN,NPTS,IPLOT,
           +                NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
      C
      C     AS ABOVE, SEE IF THE ENDPOINT OF THE SEARCH INTERVAL (HERE, VAR(5))
      C     YIELDS A BOUNDARY MIN SSE.  IF SO, RESET THE VARIABLES SO THAT
      C     THE RANGE IS EXTENDED UPWARDS.  OTHERWISE, FIND THE STARTING POINT
      C     FOR THE NEXT ROUND, AND GO TO PRECISION CHECK SECTION.
      C
            IF( (SSE(5) .GT. SSE(3)) .OR. (SSE(5) .GT. SSE(4)) ) GOTO 400
            IF(IPRINT .EQ. 0) GOTO 350
            PRINT 160, IROUND,ALPH(NOTE),VAR,SSE
            NOTE = NOTE + 1
            IF(NOTE .EQ. 28) NOTE = 27
            CALL CLOSE(6)
      350   IF(ISTOP .NE. 0) GOTO 650
            CALL RESET(VAR,3,1,2,3)
            CALL RESET(SSE,3,1,2,3)
            CALL RESET(AMP,3,1,2,3)
            VAR(5) = VAR(3) + (VAR(3) - VAR(1))
            IF(VAR(5) .NE. VAR(3)) GOTO 375
            ISTART = 2
            GOTO 434
      375   CALL SSESUB(VAR(5),SSE(5),AMP(5),IAMP,WORK,STAN,NPTS,IPLOT,
           +                NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
            GOTO 300
      C
      400   ISTART = 2
            IF(SSE(4) .LE. SSE(3)) ISTART = 3
      420   IF(IPRINT .EQ. 0) GOTO 430
            PRINT 160, IROUND,ALPH(NOTE),VAR,SSE
            CALL CLOSE(6)
      C
      C     PRECISION CHECKS
      C
      C     FIRST:        IF THE DIFFERENCE BETWEEN SUCCESSIVE ESTIMATES OF THE
```

```
      C          UNKNOWN VARIABLE IS LESS THAN OR EQUAL TO "ACCTHE"
      C          (USER-SUPPLIED CONSTANT) THEN EXIT.
      C
      C          THE DIFFERENCE MAY NEVER BE LESS THAN ACCTHE BECAUSE OF
      C          ROUNDING, IN THIS CASE DIFF REMAINS CONSTANT OVER ROUNDS.
      C          DIFOLD IS A CHECK ON THIS.
      C
      430   DIFF = VAR(2) - VAR(1)
            IF(DIFF .GT. ACCTHE) GOTO 432
            PRINT 431,ACCTHE
      431   FORMAT('0***** SEARCH STOPPED     DIFFERENCE BETWEEN ESTIMATES <=',
           +             G15.5)
            GOTO 700
      432   IF(DIFF .NE. DIFOLD) GOTO 440
      434   PRINT 435,VARNAM
      435   FORMAT('0*****SEARCH STOPPED*****'/
           +     ' DOUBLE PRECISION REQUIRED FOR FURTHER REFINEMENTS OF',
           +     ' ESTIMATES OF ',30A1)
            GOTO 700
      C
      C     SECOND:   IF THE SSE'S AGREE TO A USER-SUPPLIED NUMBER OF
      C                  DIGITS (IACCSE) THEN EXIT.
      C
      440   DIFOLD = DIFF
            ENCODE (14,450,ACCSE1(1))SSE(ISTART)
      450   FORMAT(1PE14.7)
            DO 550 I=(ISTART+1),(ISTART+2)
               ENCODE(14,450,ACCSE2(1))SSE(I)
               DO 500 J=12,14
                IF(ACCSE1(J) .NE. ACCSE2(J)) GOTO 600
      500       CONTINUE
               DO 550 J=2,(IACCSE + 2)
                IF(ACCSE1(J) .NE. ACCSE2(J)) GOTO 600
      550       CONTINUE
            PRINT 575,IACCSE
      575   FORMAT('0***** SEARCH STOPPED     SUM OF SQUARED ERRORS MATCH TO',
           +              I2 ' DIGITS.')
            GOTO 700
      C
      C     SET UP ARRAYS FOR THE NEXT ROUND
      600   IF(ISTOP .NE. 0) GOTO 650
            CALL RESET(VAR,ISTART,1,3,5)
            CALL RESET(SSE,ISTART,1,3,5)
            CALL RESET(AMP,ISTART,1,3,5)
            IROUND = IROUND + 1
            NOTE = 1
            GOTO 150
      C
      650   PRINT*,'***** SEARCH STOPPED    PLOTTING ARRAYS FILLED *****'
      700   THETA = VAR(ISTART+1)
            SSETHE = SSE(ISTART+1)
            AMPTHE = AMP(ISTART+1)
      C
            IF(IPLOT .EQ. 0) RETURN
```

```
      C
      C      ORDER THE VARIABLE ESTIMATES SO THAT A LINE CAN BE DRAWN BETWEEN
      C      DATA POINTS ON THE GRAPH,  SHOWING BEHAVIOR OF SSE'S.
      C
      800    FORMAT ('1')
             DO 900 I=1,(NPOINT - 1)
                MIN = I
                DO 850 J=I,NPOINT
                 IF(VARPLT(J) .LT. VARPLT(MIN)) MIN = J
      850        CONTINUE
                S1 = VARPLT(I)
                VARPLT(I) = VARPLT(MIN)
                VARPLT(MIN) = S1
                S1 = SSEPLT(I)
                SSEPLT(I) = SSEPLT(MIN)
      900       SSEPLT(MIN) = S1
      C
             IF(IPLOT .EQ. 2) GOTO 950
      C
             PRINT 800
      C      REGULAR PLOT OF SSE VS VARIABLE ESTIMATES.
             CALL DOPLOT(XSIZE,YSIZE,VARPLT,SSEPLT,NPOINT,VARNAM,NCHAR,'SSE',
            +                    3,0,'.',1,1)
      C
             IF(IPLOT .EQ. 1) RETURN
      C
      950    PRINT 800
      C      SEMILOG PLOT OF SSE VS VARIABLE ESTIMATES.
             CALL PLTLGY(XSIZE,YSIZE,VARPLT,SSEPLT,NPOINT,VARNAM,NCHAR,'SSE',
            +                    3,0,'.',1)
             RETURN
             END
```

```
      C
      C      SUBROUTINE SSESUB
      C
      C      PURPOSE
      C         CALLED BY SEARCH TO FIND AMPLITUDE RATIO(AMP) AND SSE FOR A
      C         GIVEN VALUE OF THE UNKNOWN PARAMETER(VAR)
      C
             SUBROUTINE SSESUB(VAR,SSE,AMP,IAMP,WORK,STAN,NPTS,IPLOT,
            +              NPOINT,VARPLT,SSEPLT,NPLTPT,ISTOP)
             DIMENSION WORK(NPTS),STAN(NPTS),VARPLT(NPLTPT),SSEPLT(NPLTPT)
      C
             CALL FUNCT(VAR,WORK)
      C
             A = 1.
             IF(IAMP .EQ. 0.) GOTO 200
      C      FIND AMPLITUDE RATIO
             SUMSQ = 0.
             SUMX = 0.
             DO 100 I=1,NPTS
                SUMX = SUMX + STAN(I)*WORK(I)
      100       SUMSQ = SUMSQ + WORK(I)*WORK(I)
             AMP = SUMX/SUMSQ
             A = AMP
      C
      C      FIND SSE
      200    SSE = 0.
             DO 300 I=1,NPTS
                ERROR = STAN(I) - A*WORK(I)
      300       SSE = SSE + ERROR**2
      C
      C      FILL PLOTTING ARRAYS IF NECESSARY
      C
             IF(IPLOT .EQ. 0) RETURN
             NPOINT = NPOINT + 1
             VARPLT(NPOINT) = VAR
             SSEPLT(NPOINT) = SSE
      400    IF(NPOINT .GE. (NPLTPT - 1) ) ISTOP = 1
      C      IF ISTOP = 1, ARRAYS ARE FILLED AND SEARCH STOPS UPON RETURN.
      C
             RETURN
             END
```

```
C
C     SUBROUTINE RESET
C
C     PURPOSE
C        CALLED BY SEARCH TO REASSIGN ELEMENTS OF ARRAYS.
C
      SUBROUTINE RESET(A,I,J,K,L)
      DIMENSION A(5)
      X = A(I)
      Y = A(I+1)
      Z = A(I+2)
      A(J) = X
      A(K) = Y
      A(L) = Z
      RETURN
      END
```